


# HP-85 Roses

## Rose Curve

```
100 CLEAR @ DEG
110 T$=CHR$(16)
120 DISP "Rose curve:"
130 DISP " r=asin(k";T$;" )"
140 DISP "where:"
150 DISP " k=p/q is a ";
160 DISP "rational number"
170 DISP " d";T$;"=step size";
180 DISP " of ";T$;" ";
190 DISP "in degrees"
200 DISP @ DISP "a,p,q,d";T$;
210 INPUT A,P,Q,D
220 IF A>1 THEN A=1
230 IF D<EPS THEN D=1
240 P=IP(P) @ Q=IP(Q)
250 IF P<1 OR Q<1 THEN 200
260 F=2-P*Q MOD 2
270 C=180*Q*F @ K=P/Q
280 SCALE -1.4,1.4,-1.05,1.05
290 MOVE 0,0
300 FOR T=D TO C STEP D
310 R=A*SIN(K*T)
320 X=R*COS(T) @ Y=R*SIN(T)
330 DRAW X,Y
340 NEXT T
350 DRAW 0,0
900 END
```



## Maurer Rose

```
100 CLEAR @ DEG
110 T$=CHR$(16)
120 DISP "Rose curve:"
130 DISP " r=sin(n";T$;" )"
140 DISP "Maurer rose:"
150 DISP " polygonal curve ";
160 DISP "with vertices"
170 DISP " on a rose (";
180 DISP T$;"=0,d,2d,...,360d)"
190 DISP @ DISP "n,d";
200 INPUT N,D
210 N=IP(N) @ D=IP(D)
220 IF N<1 OR D<1 THEN 190
230 T0=0 @ C=0 @ GCLEAR
240 SCALE -1.4,1.4,-1.05,1.05
250 T=T0 MOD 360
260 R=SIN(N*T)
270 X=R*COS(T) @ Y=R*SIN(T)
280 MOVE X,Y
290 T=(T+D) MOD 360
300 R=SIN(N*T)
310 X=R*COS(T) @ Y=R*SIN(T)
320 DRAW X,Y
330 C=C+1 @ IF T#T0 THEN 290
340 T0=T0+1 @ IF C<360 THEN 250
900 END
```

